Mathematical Sciences Individual Project Report

Course tutors: **A L Hunt and D Quibell**

Project Supervisor: **Kevin Parrot**

A report submitted in partial fulfilment of the
University of Greenwich
Research Methods and Project Course (MATH 1048)

UNIVERSITY of GREENWICH

HOW LONG CAN SOMEONE SURVIVE IN A FIRE

Salaah Amin

9 April 2015

1. ABSTRACT

Computational fluid dynamic (CFD) models are used to research the behaviour of fire. They are ideal for their accuracy and flexibility in including a number of conditions. However, CFD models are complex and require prerequisite knowledge to perform. They require large amounts of computing power to run (Icove, 2011) and for a researcher, it may not be viable to use a CFD model at home.

The report follows the implementation of a two zone model produced on Microsoft Excel using VBA. It does not require large amounts of computational power. Although as a zone model, it is incapable of using complex shapes, it is able to provide a reasonable degree of accuracy (Tavelli, Rota, & Derudi, 2014). It calculates the behaviour of fire by separating a room to a hot and cool layer. The program is capable of calculating the depth, height, mass, enthalpy and temperature of a hot layer. It also explorers carbon monoxide produced by the fire. Using this information, it is able to estimate how long a person can survive in a fire.

The testing of the program has shown that it is able to produce reasonably accurate results when exploring smoke related deaths. However, it has not been tested for heat related death accuracy.

2. INTRODUCTION

This report will follow the implementation of a fire modelling program produced on Microsoft Excel. The program utilises a two zone model to explore the behaviour of fire in a closed environment. It will then calculate how long an average person can survive in a closed environment with a live fire.

Computational fluid dynamic (CFD) models are used to simulate the behaviours of fires. These models are more accurate in comparison to the results produced by a two zone model. However, CFDs require a large amount of time to run (Icove, 2011) and require large amounts of computational power (Fire Safety Engineering Group, 2013). The implemented program will not require large amounts of computational power, thus the hardware requirements will not be as expensive. The biggest advantage of the program would that simulations can be relatively quickly. This makes it a viable option for when results need to be produced quickly.

3. HUMAN SURVIVABILITY

An experiment on human survivability was conducted in 2010 where the temperature was exceeding 200^{o}F. Results had shown that humans were most susceptible to respiratory burns from heat first, toxic smoke second and then humidity (Marsar, 2010). It is also known that the upper limit of human temperature tenability is approximately 212^{o}F (National Fire Protection Association, 1986).

Although it can require temperatures of near 200^{o}F to kill, the human skin begins to feel pain at 111^{o}F and at 162^{o}F, skin is instantly destroyed (NIST, 2013). However, how long a person can survive at a certain temperature would vary from person to person. The program will therefore assume that a temperature of 212^{o}F will result in instantaneous death.

The toxins produced from fire contain a mixture of products, one of which is carbon monoxide (CO). Appendix 14 is a graph of appendix 11. It illustrates how the dangers with carbon monoxide very quickly increase as concentration increases. Carbon monoxide poisons a human by attaching itself to the haemoglobin in the red blood cells to become carboxyhaemoglobin COHb. This stops the haemoglobin from carrying oxygen to the major organs and muscles. A prolonged time without oxygen can cause asphyxiations and eventually death. However, appendix 14 shows that a lot of time is not required to kill a person given the correct concentration. The program will investigate the effects of CO as more fire deaths occur from CO than any other toxin from a fire (Berkman & Hay, 2002).

4. TWO ZONE MODEL MECHANICS

A two zone fire model uses two control volumes to describe a compartment – a hot upper layer and a cooler lower layer (Jones, 2001). The zone model represents the convection of heat and smoke to the hotter layer. Likewise, it also represents the relatively cooler and cleaner air at the lower layer.

Appendix 1 illustrates that, a two zone model does not model fire to its exact dynamic shapes. However, through the use of simple shapes it is able to produce reasonable results when modelling an empty room (Tavelli, Rota, & Derudi, 2014).

The mechanics behind a two zone model assumes that pressure remains constant. It assumes that there is sufficient oxygen to sustain free burning and only mass from the fire plume move to the hot layer (Fire Safety Engineering Group, 2013). A zone model simplifies fire so that it possible to make calculations through the use of these assumptions.

In a two zone model, mass and energy travels through the plume to the hot upper layer instantaneously. As there are only two layers, there is not a gradual change in the heat, toxicity and mass as energy travels to the upper layer. This makes zone models ideal for smaller rooms. Considering an intermediate region where the diffusion of gasses take place, in a smaller room this region would be relatively small. Therefore, the two zone model would be producing a fire in a way which is more similar to a real fire.

5. PROJECT DESIGN AND CALCULATIONS

5.1 PROGRAM LAYOUT AND INITIAL CONDITIONS

The program allows the input of a number of variables which dictate the behaviour of the fire. These include the dimensions of the room, variables affecting time, the height of the human, and carbon monoxide concentration. The project design (appendix 9) shows a full list of inputs. The program analyses the behaviour of fire in a closed room with a slopped roof. The calculations produced include: the temperature, depth, height, mass and enthalpy in the hot layer. It is also able to calculate the concentration of carbon monoxide in the subject's body. Coupled with the fraction of incapacitating dosage equation, it is capable of calculating how much of a lethal amount of CO has been inhaled.

The program can be run via Microsoft Excel using ActiveX controls. The workbook includes four sheets. Appendix 2 shows sheet 1 which presents the user with a list of variables which they will need to assign values for. The program will then be able to calculate how long a person could survive exploring explicitly heat and carbon monoxide as causes of death. Sheet 1 includes a tick box which when ticked will allow the option to add the concentration for CO. This gives the user the option to ignore effects of carbon monoxide. Usually, a separate model is used to calculate the effects of gases. The implemented model assumes that the concentration of the toxin will remain constant. Whereas, other models include carbon dioxide (CO_2) in the calculations. The concentration of carbon dioxide in a person can dictate their breathing rate. This would change the rate of absorption of toxins. Thus, the rate at which toxins is absorbed is not constant. The implemented program requires the user to input a constant value of CO which does not factor in CO_2 in its calculations.

Unlike the model in appendix 1, the program includes the formulation of a slopped roof. This is to allow the programme to be applicable to rooms without a flat ceiling, such as attics.

The program models uses the energy output as $2t + 1$. A zone model assumes that there is a constant supply of oxygen. It also assumes that the energy radiated from the flames remains constant throughout the fire (Galea, 2014). Therefore, a flashover does not take place, and so the energy output can never decrease. $2t+1$ can be used to approximate the heat release rate at the early stages of a fire.

Appendix 3 shows sheet 2 which is used to store the results from the model. Sheet 4 includes a graph which provides a visual representation of the results of the fire. As shown on appendix 4, a scrollbar below the graph allows the user to scroll and view how this changes results over the duration of the fire. Furthermore, through the use of tick boxes, the user can choose the information they wish to view. This also allows the possibility to easily stack different graphs on top of each other for an easier comparison of results.

Essentially, two sheets provide a set of results in such a manner that numerous graphs do not need to printed and compared side by side. This program has been tailored to produce results in a simple and comprehensive manner. Due to the simplicity of the program's layout, its uses can extend to being a teaching aid. ActiveX controls help in making the program user friendly. The controls would seem familiar to the user as they would have seen similar controls on the internet. The familiarity would make the program more user friendly.

5.2 CALCUATIONS IN THE HOT LAYER

The program has been implemented using Visual Basic. The first step in the program is to declare the variables and import the user inputted variables from sheet 1. The program then calculates the constants shown in appendix 5. These constants lead to calculations which determine the changes of enthalpy and mass in the hot layer.

One of the constants calculated is the total number of iterations. Results are stored in arrays to ensure information is not lost. This calculation is used to define the size of the arrays. The calculation for this is total time / time step and the value is stored as M.

The rate at which smoke occupies a space is dependent on its shape. The total volume of the roof is calculated and stored. This is to ensure that when calculating the height of the fire, the program is able to differentiate the shapes in the room.

Initially, at time = 0, the value for temperature, mass, depth and enthalpy in the hot layer is 0. The height of the hot layer is equal to the total height of the building. As there is no fire at time = 0 and thus no hot layer, the height of the hot layer must equal to the height of the building. It cannot equal to 0 as that would imply that the building is completely filled with smoke.

Changes in mass ($\dot{m}$) and enthalpy ($\dot{e}$) can be calculated using the formulae in appendix 6. Using these, Euler's method can then be used to calculate the new values for enthalpy (Eh) and mass (Mh) in the hot layer. Euler's explicit method uses the idea that $Y_{i+1} = Y_i + \dot{y}$ where i represents the iteration number. As illustrated in program's VBA script (appendix 10), using this concept and a while loop, i is incremented from 0 to M where at each iteration, results are produced.

Using the values for $\dot{m}$ and $\dot{e}$, the depths and temperatures of the hot layer can be calculated as shown in appendix 7 at each time interval.

The program calculates the height of the hot layer by separating the slopped roof from the remainder of the house. Firstly, it calculates the volume of the smoke by multiplying the depth of the hot layer by the area. Assuming this value is not greater than the maximum volume of the roof, the program calculates the height of the hot layer to be:

$\mathrm{Zh} = (\mathrm{hRoom} + \mathrm{hRoof} - \mathrm{hF}) - 3 \times \mathrm{volHot}/\mathrm{A}$, where Zh = height of the hot layer; hRoom = height of the room; hRoof = height of the roof; hF = height of the fire from the base; volHot = volume of smoke and A = area.

This equation uses the concept that the volume of a pyramid is $\frac{1}{3}\mathrm{lwh}$. The height of the hot layer (Zh) given by height of the room – depth of the hot layer (Galea, 2014). Using the depth to calculate a volume for the smoke, a comparison can be made with the capacity of the roof. The rate is multiplied by 3 as the volume of a pyramid is a third of the volume of a room with the same length, width and height.

If the volume of the smoke is greater than total volume of the roof, then an alternative equation is used. The height is calculated as:

$Zh = (hRoom + hRoof - hF) - hRoof - \frac{volHot - maxVolRoof}{A}$, where maxVolRoof = total volume of the roof. This formulation takes the original formula for calculating the height of the fire (Zh) and firstly subtracts the height of the roof as it is completely occupied by smoke. It then subtracts the remaining volume of the smoke and divides by A to produce a one dimensional value.

Running the program shows that the rate at which smoke fills up the roof is faster than the rate at which the remainder of the building is filled. This is illustrated by appendix 8 as the rate changes significantly after 8 seconds. This shows that the program is able to detach the roof from the building and produce results based on different shapes.

5.3 FIRE SMOKE

The program explicitly concentrates on the effects of carbon monoxide on survivability. It calculates the concentration of carboxyhaemoglobin in blood. It also calculates the fractional incapacitating dose of carbon monoxide.

The program assumes that the concentration of carbon monoxide in the atmosphere remains constant. It also assumes that the volume of air breathed per minute (l/min) is 25 litres. This would be similar to the RMW of a person undergoing light activity (Purser, n.d).

Using RMW = 25 and the carbon monoxide concentration inputted by the user, the concentration of carboxyhemoglobin in a person's blood can be calculated. The concentration of COHb can be calculated using: $\mathrm{COHb} = (3.317 \times 10^{-5})(\mathrm{ppm\ CO}^{1.036})(\mathrm{RMW})(\mathrm{t})$ (Purser, n.d). The concentration is dependent on a person's breathing rate. Over the course of a fire, the concentration of carbon monoxide in the blood will begin to increase. As more red blood cells lose their ability to carry oxygen, the person would start to breathe more rapidly. This would increase RMW, and thus increase the rate at which they are absorbing carbon monoxide. Using RMW as a constant 25 does not produce exact results. RMW would vary from person to person and would change as more toxins from the fire is absorbed. A person panicking would absorb carbon monoxide faster in comparison to someone who is not. Therefore, as well as considering a person's physical state when calculating RMW, a person's mental state needs to be considered.

The program calculates the fraction of incapacitating dose using the concentrated of CO inputted by the user as shown in appendix 15. This calculation explains how near a person is to having a lethal amount of caroxyhemglobin in their blood. As the concentration of COHb in the blood increases, the equation shows how close it is to the lethal amount. The program assumes that a 30% concentration of COHb is lethal.

5.4 EFFECTS OF FIRE ON A PERSON

The model considers three variables when calculating a person survivability. Firstly, it calculates if the hot layer has descended to the height of the person. Once that height has been achieved, the results produced by the program is highlighted in orange. This provides a visual aid.

Secondly, it considers the temperature of the hot layer. As the upper limit of human tenability for heat is 212^{o}F, if the temperature exceeds this, it assumes instantaneous death. The program then highlights results produced for the temperature as red. This indicates that the temperature at that particular time would have caused the death.

Similarly, when the hot layer reaches the subject, the COHb concentration is evaluated. If it is of a lethal amount, the program declares the the person dead. This again, is illustrated by the red highlighting of the COHb results.

Although the fraction of incapacitating dose increases as the concentration of COHb increases, it is not used to declare the subject dead. The equation represents how close a person is to the lethal dose of CO. Only at 100% is it certain that a person would die from the toxin. However, this would result in the same stopping point as COHb at 30%. This would essentially be a repetition of information. Instead, this calculation can be used to observe how fast a subject's blood is absorbing carbon monoxide.

5. TESTING

The program was run in different scenarios to investigate how accurate the results to in comparison to the information provided in appendix 11. Simulations were run with concentrations of carbon monoxide at 400 ppm, 800 ppm, 1600 ppm, 3200 ppm, 6400 ppm and 12800 pm. Appendix 12 shows the input variables for the simulations. All values except the concentration of carbon monoxide remained constant throughout the simulations.

Appendix 13 shows the results of the simulation. The graph show the time it took for the concentrations of COHb to reach 30% and 100%. When the concentration of COHb reaches 30% in the blood, it can lead to incapacitation (Purser, n.d). Hence, this value is considered as a lethal concentration of COHb. In general, the results from the program suggest that a person would die quicker than what had been indicated by appendix 11.

In the simulation, the time at which the concentration of carbon monoxide reaches 100% was also included as this provides a natural upper bound. As there is no haemoglobin to carry blood at this point, death is considered a certainty. As expected, the simulation shows the time until a person dies at a concentration of 100% COHb is greater than COHb concentrations of 30%. However, it was still less that the values from appendix 11.

The simulation showed that the time before death is relatively varied between each of the three calculations initially. Although, as the concentration of carbon monoxide increases, the variation between the times decrease.

The graph would suggest that the model is not reliable at calculating the human survivability at lower concentrations of carbon monoxide. However, the volume of carbon monoxide inhaled would vary depending on a person's breathing rate. The implemented program used a 25 litres as the volume of air breathed each minute. This value is given to a person undergoing light activity (Purser, n.d). Therefore, there would be a slightly elevated breathing rate. The shape of the graph in appendix 12 suggests that the respiratory rate used to calculate the values in appendix 11 must be lower.

In a scenario where concentrations of carbon monoxide is high, it will become difficult to breathe for everyone. Therefore, naturally every person's breathing rate would increase regardless what it was initially. This would increase the concentration of COHb in the blood, and thus would decrease the time before death. This fact supports the assumption that the respiratory rate used to calculate the values in appendix 11 must be lower.

Considering this assumption on appendix 11 and the simulations results, it is evident that the model provides a good degree of accuracy in calculating carbon monoxide related deaths. However, the ability to input the respiratory rate would further improve the accuracy of the model.

In order to calculate the program's accuracy in calculating heat related deaths, the problem will need to be tested against a CFD or a model specialised in calculated heat. As a CFD model provides a more accurate result in comparison to a zone mode, a comparison between the two results is essential. This could lead into investigations as to how accurate the program is in comparison to CFDs.

6. SUMMARY AND CONCLUSION

The implemented program fulfils its aims in providing a relatively fast way in simulating fires whilst maintaining a degree of accuracy. The program has produced successful results when simulating carbon monoxide related deaths. However, deaths caused by high temperatures could not be tested.

The program can be used as a learning tool for understanding the behaviour of fire in a closed environment. The use of ActiveX controls along with VBA allows for the use of a dynamic graph. This graph provide an interactive visual aid which can assist in learning. The program would be useful for someone studying fire at home.

The program currently uses a pre-set linear equation to calculate heat release. In order to improve accuracy and usefulness, this needs to become a manual input. The program can potentially be improved to provide more analysis on the toxins within the smoke. By adding calculations for concentrations of calculating hydrogen cyanide (HCN) and carbon dioxide (CO2) the program can calculate the effects of toxins in general. $\mathrm{F_{IN}} = \left(\mathrm{F_{I_{CO}}} + \mathrm{F_{I_{CN}}}\right) \times \mathrm{VCO_2} + \mathrm{F_{I_{CO_2}}}$ (Purser, n.d) can be used to predict when the symptoms of the toxins will show. This equation includes the major toxins found in smoke where $\mathrm{F_{I_z}}$ = is the fraction of incapacitating dose of the toxic z and $\mathrm{VCO_2}$ = multiplication factor for $\mathrm{CO_2}$ induced hyperventilation (Purser, n.d).

Due to the capabilities of zone models, the program can be limited. Appendix 1 shows that a zone model does not consider the direction of the wind. If a window was open, the way in which fire behaves would be very different. As well as the heat travelling towards the ceiling, some of it will escape. As a zone model assumes that all energy is conserved, it will not provide accurate results where energy is lost. Similarly, this model cannot be used to explore behaviours of a fire outside a closed room. A zone model calculates fire using layer heights and as the ceiling being the highest point. A bin fire outdoors does not have a ceiling to use as the upper limit as the layer height. Therefore, the program is restricted to fires in a closed roof.

In conclusion, the program provides a good basis in understanding fire within a closed compartment. Results are accurate and easy to produce. It offers potential to be improved providing the addition of more inputs and generally using less constants for its calculations.

7. REFERENCES

Berkman, B., & Hay, A. (2002). *CO RX: A Safety Prescription* (4th ed.). Brooklyn, NY: FDNY WNYF.

Curkeet, R. (2011). *Wood Combustion Basics.* Intertek.

Fire Safety Engineering Group. (2013). *Smartfire Hardware Requirements*.

FSEG. (n.d.).

Galea, E. (2014). *Principles and Practice of Fire Modelling.* London: FSEG.

Greiner, T. H. (1997). *Carbon Monoxide Concentrations: Table (AEN-172).* Department of Agricultural and Biosystems Engineering, Iowa State University.

Icove, D. J. (2011). The Fire Modeling Process. Tennessee: The University of Tennessee Knoxville.

Jones, W. W. (2001). *State of the Art in Zone Modeling of Fires.* Gaithersburg: National Institute of Standards and Technology. Retrieved from

Kane. (n. d). *Carbon Monoxide.* Kane International Limited.

Marsar, S. (2010). *Survivabiity Profiling: How Long can Victims Survive in a Fire?* Fire Engineering.

National Fire Protection Association. (1986). *Instructor's Guide.*

NIST. (2013). *Fire Dynamics.*

Purser, D. A. (n.d). *Modelling Toxic and Physical Hazard in Fire.* Huntingdon: International Association for Fire Safety Science.

Shoebridge, T. (2012). *Carbon Monoxide & Hydrogen Cyanide Make Today's Fires More Dangerous.* Tulsa: Firefighter Nation.

Sjostrom, E. (2005). *Wood Chemistry: Fundamentals and Applications* (Second ed.). San Diego: Academic Press. Retrieved from

Tavelli, S., Rota, R., & Derudi, M. (2014). *A Critical Comparison Between CFD and Zone Models for the Consequence Analysis of Fires in Congested Environments.* The Italian Association of Chemical Engineering.

US Environmental Protection Agency. (1991). *Air quality criteria for carbon monoxide.* Washington, DC: Office of Research and Development.

Wolchover, N. (2012). *What Are the Limits of Human Survival?* LiveScience.

8. APPENDICES

Appendix 1: Zone Model Representation of a Compartment Fire (Fire Safety Engineering Group, 2013)



Appendix 2: Sheet 1 from the Implemented Program



Variable	Value
Length of the Room (m)	4
Width of the Room (m)	5
Height of the Room (ignoring the roof) (m)	3
Height of the Roof (m)	3
Distance Between the Floor and the Base of the Fire (m)	0
Ambient Temperature of the Room	293
Ambient Pressure of the Room (kg/m^3)	1.18
Specific Heat Constant	1.004
K	0.076
Time Step (s)	0.5
End Time (s)	500
Fraction of Heat Radiated from the Fire	0.35
Lump Amount of Energy Lost	0.6
Height of Subject (m)	2
Concentration of Carbon Monoxide (ppm)	1

Appendix 3: Sheet 2 from the Implemented Program

t	Qdot	mH	dH	eH	z	tH	COHb%	FICO
0.00	0.10	0	0	0	6	293	0.00%	0.00%
0.50	1.10	0.60539	0.02566	178.129	5.92303	293.066	0.04%	0.14%
1.00	2.10	1.92311	0.08156	566.204	5.75533	293.249	0.08%	0.28%
1.50	3.10	3.48137	0.14771	1025.44	5.55688	293.378	0.12%	0.41%
2.00	4.10	5.15487	0.2188	1518.98	5.34361	293.495	0.17%	0.55%
2.50	5.10	6.87583	0.29195	2026.88	5.12414	293.608	0.21%	0.69%
3.00	6.10	8.6017	0.36538	2536.62	4.90387	293.723	0.25%	0.83%
3.50	7.10	10.3044	0.43788	3039.93	4.68637	293.839	0.29%	0.97%
4.00	8.10	11.965	0.50865	3531.28	4.47405	293.959	0.33%	1.11%
4.50	9.10	13.5711	0.57717	4006.99	4.26848	294.083	0.37%	1.24%
5.00	10.10	15.1149	0.64311	4464.77	4.07067	294.212	0.41%	1.38%
5.50	11.10	16.5917	0.70627	4903.26	3.88119	294.347	0.46%	1.52%
6.00	12.10	17.9994	0.76656	5321.8	3.70033	294.487	0.50%	1.66%
6.50	13.10	19.3374	0.82395	5720.23	3.52815	294.634	0.54%	1.80%
7.00	14.10	20.6063	0.87847	6098.77	3.36458	294.786	0.58%	1.93%
7.50	15.10	21.8079	0.9302	6457.87	3.2094	294.946	0.62%	2.07%
8.00	16.10	22.9442	0.97922	6798.17	3.06235	295.111	0.66%	2.21%
8.50	17.10	24.0177	1.02563	7120.42	2.97437	295.284	0.70%	2.35%
9.00	18.10	25.0611	1.07083	7434.19	2.92917	295.461	0.75%	2.49%
9.50	19.10	26.0976	1.11579	7746.35	2.88421	295.639	0.79%	2.63%
10.00	20.10	27.126	1.16047	8056.52	2.83953	295.82	0.83%	2.76%
10.50	21.10	28.1452	1.20482	8364.38	2.79518	296.002	0.87%	2.90%
11.00	22.10	29.1543	1.24879	8669.65	2.75121	296.187	0.91%	3.04%
11.50	23.10	30.1522	1.29235	8972.07	2.70765	296.373	0.95%	3.18%
12.00	24.10	31.1385	1.33547	9271.43	2.66453	296.562	1.00%	3.32%
12.50	25.10	32.1123	1.37812	9567.55	2.62188	296.753	1.04%	3.46%
13.00	26.10	33.0733	1.42028	9860.27	2.57972	296.946	1.08%	3.59%
13.50	27.10	34.0208	1.46194	10149.5	2.53806	297.142	1.12%	3.73%

Notation	Definition
t	Time
Qdot	Total Rate of Energy Supplied by the Fire
mH	Mass of the Hot Layer
dH	Depth of the Hot Layer
eH	Enthalpy in the Hot Layer
z	Height of the Fire
tH	Temperature of the Hot Layer
COHb %	Concentration of CO in the Blood
FICO	Fraction of Incapacitating Dose
	The Subject is Within the Hot Layer
	The subject has died to to this value

Appendix 4: Sheet 4 from the Implemented Program



Appendix 5: Constants for Modelling Fire (Galea, 2014)

Cp = 1.004

K = 0.076

$C1 = Ta \times Pa$

$C2 = K(1 - Lr)^{1/3}$

$C3 = C2 \times Cp \times Ta$

$C4 = 1 - Ll$

Appendix 6: Calculating ṁ and ė (Galea, 2014)

$$\dot{m} = C_2 Z^{5/3} \dot{Q}^{1/3}$$

$$\dot{e} = C_4 \dot{Q} + C_3 Z^{5/3} \dot{Q}^{1/3}$$

Where ṁ represents the mass in the hot layer and ė represents the enthalpy in the hot layer.

Appendix 7: Calculating the Depth and Temperature of the Hot Layer (Galea, 2014)

$$Dh_{t+1} = \frac{Eh_{t+1}}{C1}$$

$$Th_{t+1} = \frac{Eh_{t+1}}{Mh_{t+1} C_p}$$

Where Dh represents the depth of the hot layer; Eh represents the enthalpy in the hot layer, Mh represents the mass in the hot layer and C_p represents the specific heat constant.

Appendix 8: Graph Showing Change in the Height of the Hot Layer as the Roof becomes Completely Occupied.



Appendix 9: Program Design

Step 1: Inputs
When the spreadsheet is open, there will be a set of variables that will need to be inputted by the user. Text in a cell will ask the user to input variables into column B.

Table 1: Program Input Variables

	A	B	
1	Variable	Value	
2	Length of the room (m)		rLength
3	Width of the room (m)		rWidth
4	Height of the room (ignoring the roof) (m)		hRoom
5	Height of the roof (m)		hRoof
6	Distance between the floor and the base of the fire (m)		hF
7	Ambient temperature of the room		Ta
8	Ambient pressure in the room (kg/m^3)		Pa
9	Specific heat constant	1.004	Cp
10	K	0.076	K
11	Time step (s)		Dt
12	End time (s)		tEnd
13	Fraction of heat radiated from the fire		Lr
14	Lump amount of energy lost		Ll
15	Height of Subject (m)		hSubj

Table 1 illustrates how the spreadsheet will be displayed. The fourth column in table 1 shows the variable name that will be associated with each value in VBA. This information will not be shown in the spreadsheet.
Insert a click box from the ActiveX controls. Change the label to “Add Calculations for Smoke”. If it is not clicked, hide rows 16. If it is clicked, show row 16. Cell B16 will be labelled “Concentration of Carbon Monoxide (ppm)”. It will allow the user to insert a value in the in corresponding cell in column C.

Step 2: Declaring the Variables

Declare all the input variables at the beginning of the script with comments describing what they are. This would produce a natural key. Declare all the variables as double.

```
Dim rLength As Double          'Length of room
Dim rWidth As Double           'Width of room
Dim hRoof As Double            'Height of the roof
Dim hF As Double               'Distance between the floor and the base of the fire
Dim Ta As Double               'Ambient temperature in the room
Dim Pa As Double               'Ambient pressure in the room
Dim Cp As Double               'Specific heat constant
Dim K As Double
Dim Dt As Double               'Time step
Dim tEnd As Double             'End time
Dim Lr As Double               'Fraction of heat radiated from the fire
Dim Ll As Double               'Lump amount of energy lost
```

```
Dim hSubj As Single            'Height of the subject
Dim COconc As Double           'Concentration of carbon monoxide (CO) (ppm)
```

Declare all the other variables which will be used in the script.

```
Dim A As Double                'Area of the room
Dim maxVolRoof As Double       'Maximum volume of the roof
Dim volHot As Double           'Current volume of hot layer/smoke in the room
Dim hRoom As Double            'Height of the room
Dim C1 As Double
Dim C2 As Double
Dim C3 As Double
Dim C4 As Double
Dim t As Double                'Current time
Dim M As Double                'No. of time intervals
Dim Qdot As Double             'Rate of energy output
Dim Mdot As Double             'Change of mass in the hot layer
Dim Edot As Double             'Change of enthalpy in the hot layer
Dim RMV As Double              'Volume of air inhaled per minute
```

The values for the depth, enthalpy, temperature and height of the hot layer will be stored as an array. Declare these as unrestricted arrays:

```
Dim dH() As Double             'Depth of the hot layer
Dim eH() As Double             'Enthalpy of the hot layer
Dim z() As Double              'Height of the hot layer
Dim mH() As Double             'Height of the hot layer
Dim tH() As Double             'Temperature of the hot layer
```

Another two variables which are required to calculate the effects of carbon monoxide will be stored as arrays.

```
Dim COHbBl() As Double         'COHb concentration in the blood
Dim FICO() As Double           'FED for CO
```

Declare the variable i as an integer. This will be incremented each time a set of results for the arrays has been produced.

Step 3: Reset Workbook

Results from the program are stored in sheet 2. Row 2 will include the headings, and row 3 onwards will include the results. Cells L3:Q3 will include the key. Ensure that columns L:Q are empty by deleting them. Continue to follow the steps:

As a basis, table 2 illustrates sheet 2 without any information.

Table 2: Sheet 2 Without Results

	A	B	C	D	E	F	G	H	I
1									
2	t	Qdot	mH	dH	eH	z	tH	COHB%	FICO

1. Check if cells B3 is empty. If yes, skip the remainder of these steps. Otherwise continue.
2. Go to sheet 2.
3. Select all the rows information, excluding the title row (row 2).
4. Delete all the selected row.

Step 4: Inputs and Initial Calculations

The input data from sheet 1 will be stored as variables in the program.

All of the values inputted by the user will be stored in correspondence to what the variable names is are illustrated in table 1.

Calculate:

$$\mathrm{A} = \mathrm{rWidth} \ \times \mathrm{rLength}$$
$$\mathrm{maxVolRoof} = \frac{1}{3}\mathrm{A} \times \mathrm{hRoof}$$
$$\mathrm{C1} = \mathrm{Cp} \times \mathrm{Ta} \times \mathrm{Pa} \times \mathrm{A}$$
$$\mathrm{C2} = \mathrm{K} \times (1 - \mathrm{Lr})^{\frac{1}{3}}$$
$$\mathrm{C3} = \mathrm{C2} \times \mathrm{Cp} * \mathrm{Ta}$$
$$\mathrm{C4} = 1 - \mathrm{Ll}$$
$$\mathrm{M} = \frac{\mathrm{tEnd}}{\mathrm{Dt}}$$
RMW = 25

Calculate the maximum volume of the roof and assign this value to maxVolRoof.

Step 5: Define the Size of the Arrays using M

M dictates the number of iterations there should be within the array. This is done by dividing the maximum time by the time step size.

```
ReDim dH(0 To M) As Double
ReDim eH(0 To M) As Double
ReDim z(0 To M) As Double
ReDim mH(0 To M) As Double
ReDim tH(0 To M) As Double
ReDim COHbBl(0 To M) As Double
ReDim FICO(0 To M) As Double
```

Step 6: Set and Print Initial Value

Set the initial values when time = 0 (before fire starts).

t = 0
i = 0
dH(0) = 0
mH(0) = 0
eH(0) = 0

tH(0) = 0
z(0) = hRoom + hRoof – hF
COHBBl(0) = 0
FICO(0) = 0

Set $\dot{Q}$ = 2t + 1

Select sheet 2.
In row 2, insert a title row. The title labels are: t, Qdot, mH, dH, eH, z, tH, COHB% and FICO.
Print out the initial values onto the spreadsheet under the corresponding title.

Step 7: Calculate Mass, Enthalpy, Depth and Height

Start a while loop which is continue as long as i <> M/

Calculate the change in mass and enthalpy in the hot layer:

MDot = $C_2 Z^{5/3} \dot{Q}^{1/3}$
EDot = $C_4 \dot{Q} + C_3 Z^{5/3} \dot{Q}^{1/3}$

Calculate the mass and enthalpy in the hot layer using Euler's explicit function:

$\mathrm{mH}(\mathrm{i}+1) = \mathrm{mH}(\mathrm{t}) + \mathrm{MDot}(\mathrm{t})\mathrm{dt}$
$\mathrm{eH}(\mathrm{i}+1) = \mathrm{eH}(\mathrm{t}) + \mathrm{eH}(\mathrm{t})\mathrm{dt}$

Calculate the volume of the smoke: $\mathrm{volHot} = \mathrm{dh}(\mathrm{i}+1) \times \mathrm{A}$.

If volHot < maxVolRoof then, calculate the change in the height of the hot layer. Take into consideration that the volume of a roof is $\frac{1}{3}$lwh. Thus, the rate at which the roof is occupied by some will be three times faster.

If volHot > maxVolRoof, then calculate the change in the height of the hot layer after subtracting the maximum volume of the roof.

Store the result as z(i + 1).

Step 8: Smoke Calculations

Check if the user has opted to include smoke calculations. If they have opted out, make the value of COHbB1 (i+1) and FICO (i+1) = 0.

Otherwise, calculate values for COHbB1 and FICO.

These values are percentages which represent the concentration of COHb and fraction of incapacitating dose respectively. They cannot be greater than 100%, so if the values are greater than 100%, change the value to 100%.

Print the results under the corresponding title in sheet 2. Increment i by 1, and generate Qdot.

Step 9: Human Survivability

Check if the height of the fire is less than or equal to the subject. If yes, fill the colour of all the results as orange.

Then, check if the temperature is equal to or greater than 373.15. If yes, then fill in the temperature value in the results page as red.

Check then if the value for COHbBl is equal to or greater than 0.3. If it is fill in the corresponding value for COHbBl in the results sheet as red.

Step 10: Visual Representation

In sheet 2, produce a key in the side of the results which explain what each notation is. A button in sheet 2 should take the user to sheet 4.

Sheet 4 will include a line graph which shows results for each of the variables.

The line graph should show 30 results, and the range of results should be interchangeable using a scroll bar. The scroll bar is to be linked to sheet 3 cell B3.

Table 3 illustrates the formulation of sheet 3.

Table 3: Sheet 3

	B	C	D	E	---
2	i	t	Qdot	eH	---
3	B2 + 1	B3 * Dt	Vlookup - searching C3, from sheet 2, and returning Qdot	Vlookup - searching C3, from sheet 2, and returning eH	---
4	B3 + 1	B4 * Dt	Vlookup - searching C4, from sheet 2, and returning Qdot	Vlookup - searching C4, from sheet 2, and returning eH	---
5	B4 + 1	B5 * Dt	Vlookup - searching C5, from sheet 2, and returning Qdot	Vlookup - searching C5, from sheet 2, and returning eH	---
…	---	---	---	---	---

The results will extend to 30 rows and will include all the variables as displayed in sheet 2. The line graph will retrieve its information for sheet 3.

When the scroll bar is used, it will change the value for i, and that will cause the results to be changed accordingly.

Appendix 10: VBA Code for Program

Sheet 1

```
Private Sub CommandButton1_Click()

  Call Module1.FireModel

End Sub

Private Sub SmokeOption_Click()

  If SmokeOption = True Then
    Rows("16:16").EntireRow.Hidden = False
  Else
    Rows("16:16").EntireRow.Hidden = True
  End If

End Sub
```

Sheet 2

```
Private Sub Iteration_Change()
Iteration.Max = Sheet1.Range("C12") / Sheet1.Range("C11")
LinkedCell = s
End Sub
```

Sheet 4

```
Private Sub COHb_Click()

  If COHb = True Then
    Sheets(3).Range("J:J").EntireColumn.Hidden = False
  Else
    Sheets(3).Range("J:J").EntireColumn.Hidden = True
  End If

End Sub

Private Sub dH_Click()

  If dH = True Then
    Sheets(3).Range("F:F").EntireColumn.Hidden = False
  Else
    Sheets(3).Range("F:F").EntireColumn.Hidden = True
  End If

End Sub

Private Sub eH_Click()
```

```
    If eH = True Then
        Sheets(3).Range("G:G").EntireColumn.Hidden = False
    Else
        Sheets(3).Range("G:G").EntireColumn.Hidden = True
    End If

End Sub

Private Sub FICO_Click()

    If FICO = True Then
        Sheets(3).Range("K:K").EntireColumn.Hidden = False
    Else
        Sheets(3).Range("K:K").EntireColumn.Hidden = True
    End If

End Sub

Private Sub mH_Click()

    If mH = True Then
        Sheets(3).Range("E:E").EntireColumn.Hidden = False
    Else
        Sheets(3).Range("E:E").EntireColumn.Hidden = True
    End If

End Sub

Private Sub Qdot_Click()

    If Qdot = True Then
        Sheets(3).Range("D:D").EntireColumn.Hidden = False
    Else
        Sheets(3).Range("D:D").EntireColumn.Hidden = True
    End If

End Sub

Private Sub tH_Click()

    If tH = True Then
        Sheets(3).Range("I:I").EntireColumn.Hidden = False
    Else
        Sheets(3).Range("I:I").EntireColumn.Hidden = True
    End If

End Sub

Private Sub z_Click()

    If z = True Then
```

```
    Sheets(3).Range("H:H").EntireColumn.Hidden = False
  Else
    Sheets(3).Range("H:H").EntireColumn.Hidden = True
  End If

End Sub
```

<u>Module 1</u>

```
Sub FireModel()

  'Declaring input variables

  Dim rLength As Double   'Length of room
  Dim rWidth As Double    'Width of room
  Dim hRoof As Double     'Height of the roof
  Dim hF As Double        'Distance between the floor and the base of the fire
  Dim Ta As Double        'Ambient temperature in the room
  Dim Pa As Double        'Ambient pressure in the room
  Dim Cp As Double        'Specific heat constant
  Dim K As Double
  Dim Dt As Double        'Time step
  Dim tEnd As Double      'End time
  Dim Lr As Double        'Fraction of heat radiated from the fire
  Dim Ll As Double        'Lump amount of energy lost
  Dim hSubj As Single     'Height of the subject
  Dim COconc As Double    'Concentration of carbon monoxide (CO) (ppm)


  'Declaring all the other variables

  Dim A As Double         'Area of the room
  Dim maxVolRoof As Double 'Maximum volume of the roof
  Dim volHot As Double    'Current volume of hot layer/smoke in the room
  Dim hRoom As Double     'Height of the room
  Dim C1 As Double
  Dim C2 As Double
  Dim C3 As Double
  Dim C4 As Double
  Dim t As Double         'Current time
  Dim M As Double         'No. of time intervals
  Dim Qdot As Double      'Rate of energy output
  Dim Mdot As Double      'Change of mass in the hot layer
  Dim Edot As Double      'Change of enthalpy in the hot layer
  Dim i As Integer        'Array value
  Dim dH() As Double      'Depth of the hot layer
  Dim eH() As Double      'Enthalpy of the hot layer
  Dim z() As Double       'Height of the hot layer
  Dim mH() As Double      'Height of the hot layer
  Dim tH() As Double      'Temperature of the hot layer
  Dim COHbBl() As Double  'COHb concentration in the blood
  Dim FICO() As Double    'FED for CO
```

```
Dim RMV As Double       'Volume of air inhaled per minute


'NEED TO PRODUCE A DROP DOWN MENU FOR QDOT

'Deletes the old data from sheet 2

Sheets(2).Select

Range("L:Q").EntireColumn.Delete

If Range("B3") <> "" Then
   r = Range(Range("B3"), Range("B3").End(xlDown)).Rows.Count
   Range(Range("B3"), Cells(r + 2, "K")).Delete
End If

Sheets(1).Select

'Inputs imported from sheet 1

rLength = Range("C2")
rWidth = Range("C3")
hRoom = Range("C4")
hRoof = Range("C5")
hF = Range("C6")
Ta = Range("C7")
Pa = Range("C8")
Cp = Range("C9")
K = Range("C10")
Dt = Range("C11")
tEnd = Range("C12")
Lr = Range("C13")
Ll = Range("C14")
hSubj = Range("C15")

'Inputs and inital smoke calculations

If Sheet1.SmokeOption = True Then
   COconc = Range("C16")
   RMW = 25
Else: COconc = 0
End If

A = rWidth * rLength               'Floor frea
maxVolRoof = (1 / 3) * A * hRoof


C1 = Cp * Ta * Pa * A
C2 = K * (1 - Lr) ^ (1 / 3)
C3 = C2 * Cp * Ta
C4 = 1 - Ll
```

```
M = tEnd / Dt

'Define the size the arrays using M

ReDim dH(0 To M) As Double
ReDim eH(0 To M) As Double
ReDim z(0 To M) As Double
ReDim mH(0 To M) As Double
ReDim tH(0 To M) As Double
ReDim COHbBl(0 To M) As Double
ReDim FICO(0 To M) As Double


'Sets the inital values when t = 0

t = 0
i = 0
dH(0) = 0
mH(0) = 0
eH(0) = 0
tH(0) = Ta
z(0) = hRoom + hRoof
COHbBl(0) = 0
FICO(0) = 0

Qdot = 2 * t + 0.1

'Prints the inital values when t = 0 onto sheet 2

Sheets(2).Select

Range("B3") = t
Range("C3") = Qdot
Range("D3") = mH(0)
Range("E3") = dH(0)
Range("F3") = eH(0)
Range("G3") = z(0)
Range("H3") = tH(0)
Range("I3") = COHbBl(0)
Range("J3") = FICO(0)


 While i <> M

   t = t + Dt

   Mdot = C2 * z(i) ^ (5 / 3) * Qdot ^ (1 / 3)
   Edot = C4 * Qdot + C3 * z(i) ^ (5 / 3) * Qdot ^ (1 / 3)

   mH(i + 1) = mH(i) + Mdot
```

```
eH(i + 1) = eH(i) + Edot
dH(i + 1) = (eH(i + 1)) / C1
tH(i + 1) = (eH(i + 1)) / (mH(i + 1) * Cp)

volHot = dH(i + 1) * A

If volHot <= maxVolRoof Then
   z(i + 1) = (hRoom + hRoof - hF) - 3 * volHot / A
ElseIf volHot > maxVolRoof Then
   z(i + 1) = (hRoom + hRoof - hF) - hRoof - ((volHot - maxVolRoof) / A)
End If

'Smoke calculations

If Sheet1.SmokeOption = True Then

   COHbBl(i + 1) = 3.317 * 10 ^ (-5) * COconc ^ 1.036 * RMW * t

   If COHbBl(i + 1) > 1 Then
      COHbBl(i + 1) = 1
   End If

   FICO(i + 1) = (0.00082925 * COconc ^ 1.036 * t) / 0.3

   If FICO(i + 1) > 1 Then
      FICO(i + 1) = 1
   End If

Else

   COHbBl(i + 1) = 0
   FICO(i + 1) = 0

End If

If z(i + 1) <= 0 Or dH(i + 1) > hRoom Then
   z(i + 1) = 0
End If

Qdot = 2 * t + 0.1
i = i + 1

'Print calculated values

Range("B2").End(xlDown).Offset(1, 0) = t
Range("C2").End(xlDown).Offset(1, 0) = Qdot
Range("D2").End(xlDown).Offset(1, 0) = mH(i)
Range("E2").End(xlDown).Offset(1, 0) = dH(i)
Range("F2").End(xlDown).Offset(1, 0) = eH(i)
Range("G2").End(xlDown).Offset(1, 0) = z(i)
Range("H2").End(xlDown).Offset(1, 0) = tH(i)
```

```
    Range("I2").End(xlDown).Offset(1, 0) = COHbBl(i)
    Range("J2").End(xlDown).Offset(1, 0) = FICO(i)

    'If the smoke layer reaches the subject, assess survivability

    If z(i) <= hSubj Then

        'Highlight the data in orange

        For Col = 2 To 10
          Cells(2, Col).End(xlDown).Interior.Color = RGB(255, 204, 0)
        Next Col

        'The maximum temperature a human can withstand is 373.15
        'At this point death is certain.

        If tH(i) >= 373.15 Then
            Range("H2").End(xlDown).Interior.Color = RGB(255, 0, 0)
        End If

        If Sheet1.SmokeOption = True Then
            If COHbBl(i) >= 0.3 Then
                Range("I2").End(xlDown).Interior.Color = RGB(255, 0, 0)
            End If
        End If

    End If

Wend

'Produces a key

Range("L3") = "Notation"
Range("L3").Font.Bold = True
Range("M3") = "Definition"
Range("M3").Font.Bold = True
Range("L4") = "t"
Range("M4") = "Time"
Range("L5") = "Qdot"
Range("M5") = "Total Rate of Energy Supplied by the Fire"
Range("L6") = "mH"
Range("M6") = "Mass of the Hot Layer"
Range("L7") = "dH"
Range("M7") = "Depth of the Hot Layer"
Range("L8") = "eH"
Range("M8") = "Enthalpy in the Hot Layer"
Range("L9") = "z"
Range("M9") = "Height of the Fire"
Range("L10") = "tH"
Range("M10") = "Temperature of the Hot Layer"
Range("L11") = "COHb %"
```

```
    Range("M11") = "Concentration of CO in the Blood"
    Range("L12") = "FICO"
    Range("M12") = "Fraction of Incapacitating Dose"
    Range("L3:Q14").Interior.Color = RGB(255, 255, 204)
    Range("L13").Interior.Color = RGB(255, 204, 0)
    Range("M13") = "The Subject is Within the Hot Layer"
    Range("L14").Interior.Color = RGB(255, 0, 0)
    Range("M14") = "The subject has died to to this value"

    'Producing the Ranges for the Graph

    Sheets(2).Select

    TotRows = Range(Range("B3"), Range("B3").End(xlDown)).Count + 2


    Sheets(3).Select

    For r = 3 To 33

       Cells(r, "C") = "=B" & r & "*Sheet1!$C$11"       't

       Cells(r, "D") = "=VLOOKUP(Sheet3!C" & r & ",Sheet2!B3:J" _
                & TotRows & ",2,FALSE)"                 'Qdot

       Cells(r, "E") = "=VLOOKUP(Sheet3!C" & r & ",Sheet2!B3:J" _
                & TotRows & ",3,FALSE)"                 'mH

       Cells(r, "F") = "=VLOOKUP(Sheet3!C" & r & ",Sheet2!B3:J" _
                & TotRows & ",4,FALSE)"                 'dH

       Cells(r, "G") = "=VLOOKUP(Sheet3!C" & r & ",Sheet2!B3:J" _
                & TotRows & ",5,FALSE)"                 'eH

       Cells(r, "H") = "=VLOOKUP(Sheet3!C" & r & ",Sheet2!B3:J" _
                & TotRows & ",6,FALSE)"                 'z

       Cells(r, "I") = "=VLOOKUP(Sheet3!C" & r & ",Sheet2!B3:J" _
                & TotRows & ",7,FALSE)"                 'tH

       Cells(r, "J") = "=VLOOKUP(Sheet3!C" & r & ",Sheet2!B3:J" _
                & TotRows & ",8,FALSE)"                 'COHb%

       Cells(r, "K") = "=VLOOKUP(Sheet3!C" & r & ",Sheet2!B3:J" _
                & TotRows & ",9,FALSE)"                 'FICO

    Next r

    Sheets(4).Select

End Sub
```

Appendix 11: Inhalation Times and Toxic Symptoms of Carbon Monoxide (Kane, n. d)

Concentration of CO in Air	Inhalation Times and Toxic Symptoms
9 ppm	The maximum allowable concentration for short term exposure in a living area according to ASHRAE.
35 ppm	The maximum allowable concentration for continuous exposure in any eight hour period, according to federal law.
200 ppm	Maximum concentration allowable at any time according to OSHA. Slight headaches, fatigue, dizziness, nausea after 2-3 hours.
400 ppm	Frontal headaches within 1-2 hours, life threatening after 3 hours. Maximum allowable limit in flue gas according to EPA and AGA.
800 ppm	Dizziness, nausea and convulsions within 45 minutes. Unconsciousness within 2 hours. Death within 2-3 hours.
1600 ppm	Headache, dizziness and nausea within 20 minutes. Death within 1 hour.
3200 ppm	Headache, dizziness and nausea within 5-10 minutes. Death within 30 minutes.
6400 ppm	Headache, dizziness and nausea within 1-2 minutes. Death within 10-15 minutes.
12800 ppm	Death within 1-3 minutes.

Appendix 12: Simulation Inputs

Variable	Value
Length of the Room (m)	4
Width of the Room (m)	5
Height of the Room (ignoring the roof) (m)	5
Height of the Roof (m)	0
Distance Between the Floor and the Base of the Fire (m)	0
Ambient Temperature of the Room	293
Ambient Pressure of the Room (kg/m^3)	1.18
Specific Heat Constant	1.004
K	0.076
Time Step (s)	5
End Time (s)	6000
Fraction of Heat Radiated from the Fire	0.35
Lump Amount of Energy Lost	0.6
Height of Subject (m)	1.75
Concentration of Carbon Monoxide (ppm)	12800

Appendix 13: Death Rates Achieved Using Different Models



Appendix 14: Time before Death with Different Concentrations of Carbon Monoxide



Appendix 15: Calculating the Fraction of Incapacitating Dose of CO (Purser, n.d)

$$F_{I_{CO}} = \frac{K(CO^{1.036})(t)}{D}$$

Where:	$F_{I_{CO}}$	=	fraction of incapacitating dose
	T	=	exposure time (min)(=1 in this case)
	K	=	0.00082925 for 25 l/min RMV (light activity)
	D	=	COHb concentration at incapacitation (30% for light exercise)